While Loop

```
while (condition == true) { some work; }
```

use comparison, branch to exit, and jump to continue
generally written as:

```
loop:  check condition
       branch if condition is false to exit
       some work
       j      loop
exit:
```


While Loop

```
char *str = "Hello World!";
char *s = str;
while (*s != '\0') { printf("%c", *s); s = s + 1; }
```

```
       la     $t0,str
loop:  lb     $a0,0($t0)
       beq    $a0,$0,exit
       li     $v0,11       # print character
       syscall
       addi   $t0,$t0,1
       j      loop
exit:
```

Do While Loop

```
do { some work; } while (condition == true);
```

use comparison, branch to exit, and jump to continue
generally written as:

```
loop:  some work
       check condition
       branch if condition is true to loop
exit:
```


Do While Loop

```
char *str = "Hello World!";
char *s = str;
/* note: this breaks with the empty string! */
do { printf("%c",*s); s = s + 1; } while (*s != '\0')
```

```
       la    $t0,str
       lb    $a0,0($t0)   # start the loop
loop:  li    $v0,11       # print character
       syscall
       addi  $t0,$t0,1
       lb    $a0,0($t0)
       bne   $a0,$0,loop
```

Procedures

```
int len(char *s) {
   int l;
   for (l=0; *s != '\0'; s++) l++;
   return l;
}
void reverse(char *s, char *r) {
   char *p, *t;
   int l = len(s);
   *(r+l) = '\0';
   l--;
   for (p=s+l t=r; l>=0; l--) { *t++ = *p--; }
}
void main(int) {
   char *s = "Hello World!";
   char r[100];
   reverse(s,r);
}
```



How can we do this with assembly?
* Need a way to call / return procedures
* Need a way to pass arguments
* Need a way to return a value

Procedure Call and Return

- Procedure call
  - Need to jump to the procedure
  - The return goes back to the point immediately after the call
  - Need to pass **"return address"** (instruction after call)
  - **`jal Label`**
    - $ra = PC+4 # set return address to next PC
    - PC = PC[31:28] | Label << 2 # jump to procedure
- Procedure return
  - Needs to know the return address
  - Need to jump back to the return point (after the call)
  - **`jr $ra`**
    - PC = $ra # jump back to return address

Arguments and Return Value

- Register conventions specified in PRM
  - $a0-$a3: four arguments to pass to called procedure
  - $v0-$v1: two values returned from procedure
  - $ra: return address register (set by call, used by return)
- Call chains
  - One procedure calls another, which calls another one
  - E.g., **`main`** → **`reverse`** → **`len`**
  - What happens to $ra??? (e.g., when reverse calls len)
- You must save $ra someplace!
  - Simple approach: A "free" register (can't be used by caller)
  - Leaf procedure: Doesn't make any calls. Doesn't ever save $ra.


```
        # procedure reverse($a0,$a1)
reverse:
        move    $t7,$ra         # save return address
        jal     len             # get length of source string
        blt     $v0,$0,rev_exit  # exit if empty string
        add     $t0,$a1,$v0     # null terminate target string
        sb      $0,0($t0)       # put null into end of string
        addi    $v0,$v0,-1      # decrement length (written /0)
        add     $t0,$a0,$v0     # $t0 holds p (source string)
        add     $t1,$a1,$0      # $t1 holds t (target string)
rev_loop:
        lbu     $t2,0($t0)      # get char from source string
        sb      $t2,0($t1)      # save char to target string
        addi    $t0,$t0,-1      # decrement source string ptr
        addi    $t1,$t1,1       # increment target string ptr
        addi    $v0,$v0,-1      # decrement length
        slt     $t2,$v0,$0      # is l < 0?
        beq     $t2,$0,rev_loop
rev_exit:
        move    $ra,$t7
        jr      $ra
```

```
        # procedure len($a0); returns string length in $v0
len:
        move    $t0,$a0                 # copy start ptr
len_loop:
        lbu     $t1,0($t0)              # get char
        beq     $t1,$0,len_exit         # check for null
        addi    $t0,$t0,1               # go to next character
        j       len_loop                # continue loop
len_exit:
        sub     $v0,$t0,$a0             # diff of ptrs is length
        jr      $ra
```



```
        .data
        nl:     .asciiz         "\n"
        s:      .asciiz         "Hello World!"
        r:      .space          100
        .align 2
        p:      .word           0x0
        t:      .word           0x0
        l:      .word           0x0
        .text
        # make the call to reverse
        la              $a0,s
        la              $a1,r
        jal             reverse


        see mips12.asm for the full program
```

More Procedure Call/Return

- **Caller**: The procedure that calls another one
- **Callee**: The procedure that is called by the caller

- What if callee wants to use registers?
  - Caller is also using registers!
  - If callee wants to use same registers, we must save them
  - Consider what happened with $ra in a call chain

- Register usage conventions specified by PRM
  - $t0-$t9: Temp. registers; if caller wants them, must save before call
  - $s0-$s7: Saved registers; saved by callee prior to using them

Where to save?

- Need a memory space to hold saved registers
  - Caller spills $t0-$t9 that be must saved to memory
  - Callee spills $s0-$s7 to memory, when these regs are used
  - Non-leaf caller spills $ra when making another call

- Each procedure needs locations to save registers

- In general, call-chain depth (number of called procs) is unknown, so we need to support undetermined length

- Suggestion: Use a stack. Add "stack element" onto stack for each call. The "stack element" has the locations.

Program Stack

- **Program stack**: Memory locations used by running program
  - Has space for **temporary and saved registers**
  - Has space for **local variables**, when can't all fit in registers
    - E.g., local arrays are allocated on the stack
  - Has space for **return address**
- Each procedure allocates space for these items
  - So-called "**activation frame**" (a.k.a., "activation record")
  - Purpose of locations in activation frame are known
- **Prologue** (entry point into the procedure): Allocates an activation frame on the stack
- **Epilogue** (exit point from procedure): De-allocates the activation frame, does actual return

Stack and frame pointers

- Stack pointer ($sp)
  - Keeps the address to the top of the stack
  - $29 is reserved for this purpose
  - Stack grows from high address to low
  - Typical stack operations are push/pop
- Procedure frame
  - Contains saved registers and local variables
  - "Activation record"
- Frame pointer ($fp)
  - Points to the first word of a frame
  - Offers a stable reference pointer
  - $30 is reserved for this
  - Some compilers don't use $fp





Stack and frame pointer

- Caller saves needed registers, sets up args, makes call
  - When not enough arg regs: Put arguments onto the stack
- Called procedure prologue
  - Adjust stack pointer for activation frame size to hold enough space to hold saved registers, locals, return address (non-leaf)
  - Save any saved registers to the stack
  - Save return address to the stack
- Called procedure epilogue
  - Restore return address from the stack (non-leaf)
  - Restore any saved registers from the stack
  - Return to caller